AN-757

Application Note

ADVANCED SEMICONDUCTOR DEVICES (PTY) LTD
P.O. Box 2944, Johannesburg 2000
3rd Floor, Vogas House
123 Pritchard Street/Corner Mool Street
Johannesburg
Tel. No. 23-2856

ANALOG-TO-DIGITAL CONVERSION TECHNIQUES WITH THE M6800 MICROPROCESSOR SYSTEM

Prepared by:
Don Aldridge
Applications Engineering

This application note describes several analog-to-digital conversion systems implemented with the M6800 microprocessor and external linear and digital IC's. Systems consisting of an 8- and 10-bit successive approximation approach, as well as dual ramp techniques of 3½- and 4½-digit BCD and 12-bit binary, are shown with flow diagrams, source programs and hardware schematics. System tradeoffs of the various schemes and programs for binary-to-BCD and BCD-to-7 segment code are discussed.

MOTOROLA Semiconductor Products Inc.

Analog-To-Digital Conversion Techniques with the M6800 Microprocessor System

INTRODUCTION

The MPU (microprocessing unit) is rapidly replacing both digital and analog circuitry in the industrial control environment. It provides a convenient and efficient method of handling data; controlling valves, motors and relays; and in general, supervising a complete processing machine. However, much of the information required by the MPU for the various computations necessary in the processing system may be available as analog input signals instead of digitally formatted data. These analog signals may be from a pressure transducer, thermistor or other type of sensor. Therefore, for analog data an A/D (analog-to-digital) converter must be added to the MPU system.

Although there are various methods of A/D conversion, each system can usually be divided into two sections – an analog subsystem containing the various analog functions for the A/D and a digital subsystem containing the digital functions. To add an A/D to the MPU, both of the sections may be added externally to the microprocessor in the form of a PC card, hybrid module or monolithic chip. However, only the analog subsystem of the A/D need be added to the microprocessor, since by adding a few instructions to the software, the MPU can perform the function of the digital section of the A/D converter in addition to its other tasks. Therefore, a system design that already contains an MPU and requires analog information needs only one or two additional inexpensive analog components to provide the A/D. The microprocessor software can control the analog section of the A/D, determine the digital value of the analog input from the analog section, and perform various calculations with the resulting data. In addition, the MPU can control several analog A/D sections in a timeshare mode, thus multiplexing the analog information at a digital level.

Using the MPU to perform the tasks of the digital section provides a lower cost approach to the A/D function than adding a complete A/D external to the MPU. The information presented in this note describes this technique as applied to both successive approximation (SA) A/D and dual ramp A/D. With the addition of a DAC (digital-to-analog converter), a couple of operational amplifiers, and the appropriate MPU software, an 8- or 10-bit successive approximation A/D is available. Expansion to greater accuracies is possible by modifying the software and adding the appropriate D/A converter. The technique of successive approximation A/D provides medium speed with accuracies compatible with many systems. The second technique adds an MC1405 dual ramp analog subsystem to the MPU system and, if desired, a digital display to produce a 12-15 bit binary or a 3½- or 4½-digit BCD A/D conversion with 7-segment display readout. This A/D technique has a relatively slow conversion rate but produces a converter of very high accuracy. In addition to the longer conversion time, the MPU must be totally devoted to the A/D function during the conversion period. However, if maximum speed is not required this technique of A/D allows an inexpensive and practical method of handling analog information.

Figure 1 shows the relative merits of each A/D conversion technique. Listed in this table are conversion time, accuracy and whether interrupts to the MPU are allowed during the conversion cycle.

This note describes each method listed in Figure 1 and provides the MPU software and external system hardware schematics along with an explanation of the basic A/D technique and system peculiarities. In addition, the MPU interface connections for the external A/D hardware schemes are shown. These schemes are a complete 8-bit successive approximation and a 3½-digit dual ramp A/D system, both of which externally perform the conversion and transfer the digital data into the MPU system through a PIA.

For additional information on the MC6800 MPU system or A/D systems, the appropriate data sheets or other available literature should be consulted.

MPU

The Motorola microprocessor system devices used are the MC6800 MPU, MCM6810 RAM, MCM6830 ROM and MC6820 PIA (peripheral interface adapter). The following is a brief description of the basic MPU system as it pertains to the A/D systems presented later in this application note.

The Motorola MPU system uses a 16-bit address bus and an 8-bit data bus. The 16-bit address bus provides 65,536 possible memory locations which may be either storage devices (RAM, ROM, etc.) or interface devices (PIA, etc.). The basic MPU contains two 8-bit accumulators, one 16-bit index register, a 16-bit program counter, a 16-bit stack pointer, and an 8-bit condition code register. The condition code register indicates carry, half carry, interrupt, zero, minus, and 2's complement overflow. Figure 2 shows a functional block of the MC6800 MPU.

The MPU uses 72 instructions with six addressing modes which provide 197 different operations in the MPU. A summary of each instruction and function with the appropriate addressing mode is shown in Appendix A of this note.

Characteristic	Successive Approximation			Dual Ramp			
	8-Bit Software	10-Bit Software	8-Bit Hardware	12-Bit Software	3½-Digit Software	4½-Digit Software	3½-Digit Hardware
External Hardware	8-Bit DAC Op Amp Comparator	10-Bit DAC Op Amp Comparator	8-Bit DAC SAR* Op Amp Comparator	MC1405	MC1405	MC1405	MC1405 MC14435 MC14558 (for 7-segment display)
Conversion Rate	700 μs Constant	1.25 ms Constant	60 μs for MPU, plus A/D Conversion Time	165 ms (max) Variable	60 ms (max) Variable	600 ms (max) Variable	183 μs (min) for MPU, plus A/D Conversion Time
Interrupt Capability	Allowed	Allowed	Allowed	Not Allowed	Not Allowed	Not Allowed	Allowed
Number of Memory Locations Required (Including PIA Configuration)	106	145	42	84	296	328	58
Serial Output Available	Yes	Yes	Yes	No	No	No	No

*Successive Approximation Register

FIGURE 1 – Relative Merits of A/D Conversion Techniques

The RAMs used in the system are static and contain 128 8-bit words for scratch pad memory while the ROM is mask programmable and contains 1024 8-bit words. The ROM and RAM, along with the remainder of the MPU system components, operate from a single +5 volt power supply; the address bus, data bus and PIAs are TTL compatible.

The MPU system requires a 2ϕ non-overlapping clock with a lower frequency limit of 100 kHz and an upper limit of 1 MHz.



FIGURE 2 – MPU Pin Functions

The PIA is the interface device used between the address and data buses and the analog sections of the A/D. Each PIA contains two essentially identical 8-bit interface ports. These ports (A side, B side) each contain three internal registers that include the data register which is the interface from the data bus to the A/D, the data direction register which programs each of the eight lines of the data register as either an input or an output, and the control register which, in addition to other functions, switches the data bus between the data register and the data direction register. Each port to the PIA contains two addition pins, CA1 and CA2, for interrupt capability and extra I/O lines. The functions of these lines are programmable with the remaining bits in the control register. Figure 3 shows a functional block of the MC6820 PIA.

Each PIA requires four address locations in memory. Two addresses access either of the two (A or B sides) data/data direction registers while the remaining two addresses access either of the two control registers. These addresses are decoded by the chip select and register select lines of the PIA which are connected to the MPU address bus. Selection between the data register and data direction register is made by programming a "1" or "0" in the third least significant bit of each control register . A logic "0" accesses the data direction register while a logic "1" accesses the data register.

By programming "0"s in the data direction register each corresponding line performs as an input, while "1"s in the data direction register make corresponding lines act as outputs. The eight lines may be intermixed between inputs and outputs by programming different combinations of "1"s and "0"s into the data direction register. At the beginning of the program the I/O configuration is programmed into the data direction register, after which the control register is programmed to select the data register for I/O operation.



FIGURE 3 – PIA Functions

The printouts shown for each A/D program are the source instructions for the cross assembler from the Motorola timeshare. Since the MPU contains a 16-bit address bus and an 8-bit data bus, the hexadecimal number system provides a convenient representation of these numbers. Although the assembler output is in hexadecimal, the source input may be either binary, octal, decimal or hexadecimal. A dollar sign ($) preceding a number in the source instructions indicates hexadecimal, a percent sign (%) indicates binary and an at sign (@) indicates octal. No prefix indicates the decimal number system.

Only the beginning addresses of the program and labels are shown in the source programs. These beginning addresses may be changed prior to assembling the total system program or the programs may be relocated after assembly with little or no modification.

SUCCESSIVE APPROXIMATION TECHNIQUES

General

One of the more popular methods of A/D conversion is that of successive approximation. This technique uses a DAC (digital-to-analog converter) in a feedback loop to generate a known analog signal to which the unknown analog input is compared. In addition to medium speed conversion rates, it has the advantages of providing not only a parallel digital output after the conversion is completed but also the serial output during the conversion.

Figure 4 shows the block diagram and waveform of the SA-A/D. The DAC inputs are controlled by the successive approximation register (SAR) which is, as presented here, the microprocessor. The DAC output is compared to the analog input (V_{in}) by the analog comparator and its output controls the SAR. At the start of a conversion the MSB of the DAC is turned on by the SAR, producing an output from the DAC equal to half of the full scale value. This output is compared to the analog input and if the DAC output is greater than the input unknown, the SAR turns the MSB off. However, if the DAC output is less than the input unknown, the MSB remains on. Following the trial of the MSB the next most significant bit is turned on and again the comparison is made between the DAC output and the input unknown. The same criteria exists as before and this bit is either left on or turned off. This procedure of testing each bit continues for the total number of DAC inputs (bits) in the system.

After the comparison of each bit the digital output is available immediately thus providing both the serial output as well as the parallel output at the end of the conversion. The serial output provides the MSB first, followed by the remaining bits in order. The total conversion time for the SA-A/D is the time required to turn on a bit, compare the DAC output with the input unknown and, if required, turn the bit off, multiplied by the total number of bits in the A/D system. The conversion time is hence constant and unaffected by the analog input value.

One SA-A/D shown in this note uses an 8-bit DAC (MC1408) to produce an 8-bit A/D; a second version uses a 10-bit DAC (MC3410)* to produce a 10-bit A/D. Both of these are used in conjunction with the MPU as an SAR. In addition, the MC1408 is shown with the MC14549 CMOS SAR as a convert-on-command system under control of the MPU. All of these A/Ds produce a binary output. However, by adding the appropriate software a BCD output or 7-segment-display outputs are available. Also by using a BCD-weighted DAC, the BCD output can be produced directly.

*MC3410 to be announced



FIGURE 4 – 4-Bit Successive Approximation Converter



FIGURE 5 – 8-Bit Successive Approximation A/D Flow Diagram

8-Bit SA Program

The flow chart for the 8-bit MPU A/D system is shown in Figure 5; Figures 6 and 7 show the software and the hardware external to the microprocessor. The DAC used is the MC1408L-8 which has active high inputs and a current sink output. An uncompensated MLM301A operational amplifier is used as a comparator while an externally compensated MLM301A or internally compensated MC1741 operational amplifier is used as a buffer amplifier for the input voltage. The output voltage compliance of the DAC is ±0.5 volt; if the current required by the D/A does not match that produced from the output of the buffer amplifier through R1 and R2, then the DAC output will saturate at 0.5 volt above or below ground, thus toggling the comparator. The system is calibrated by adjusting R1 for 1 volt full scale, and zero calibration is set by adjusting R3.

The first MPU instruction for the 8-bit A/D is in line 45 of Figure 6. After assembly, this instruction will be placed in memory location $0A00 as defined in the assembler directive of line 42. The assembled code for this program is relocatable in memory as long as the PIA addresses and storage addresses are unchanged. The program as shown requires 106 memory bytes. Source program lines 45 through 53 configure the PIAs for the proper input/output configuration. PIA1BD is used for various control functions between the MPU system and the external hardware. The exact configuration of this PIA is shown in lines 28 through 33 of Figure 6. PIA1AD provides the 8-bit output needed for the DAC. Lines 51 through 53 set bit 3 of the PIA control register to access the data register for the actual A/D program.

Lines 55 and 56 set the conversion finished flag, which consists of a LED on the hardware schematic, after which the program enters a loop in lines 63-65 which causes the MPU to wait until the cycle input line goes high. (This feature could be eliminated if the program was a subroutine of a larger control program.) In this case, when a conversion was to be made the control program would go to the A/D subroutine and return with the digital results. Lines 68 and 69 clear the PIA-A which is connected to the DAC inputs and an internal memory location. This memory location is used as a pointer to keep track of which bit of the DAC is currently being tested. Next the conversion finished line is reset indicating a conversion is in process and the carry bit of the condition code register is set. The memory location POINTR is then rotated right in line 79, moving the carry bit of the condition code register into the MSB of that memory location. Line 80 is a conditional branch that determines if all 8 bits of the DAC have been tested. After nine rotations of POINTR the carry bit will again be set indicating all 8 bits have been compared.

Program lines 81 through 83 load the previous DAC value into an accumulator and the next DAC bit is turned on for the comparator test. An 8 μs delay produced by the NOP instruction of lines 87 through 90 allows the DAC and comparator to settle to a final value before the comparator test of lines 91 and 92. At this point if the comparator was high the Yes loop is executed, which generates a simulated clock pulse and a serial output "1". If the comparator was low, lines 95 through 101 are executed, resetting the bit under test and generating a simulated clock pulse and a serial output of "0". The three NOP instructions of the Yes loop equalize the execution time between the high and low comparator loops. After completion of either the high or low comparator loop, the A accumulator which contains the new digital number is stored in PIA1AD and in a RAM memory location labeled ANS. Then the next bit of the DAC is tested in the same manner and this procedure is continued until all eight DAC inputs have been tested. When this has occurred the program returns to line 55 where the conversion finished flag is "set" and the MPU awaits the next cycle input from PIA1BD.

The total conversion time is 700 μs for the 8-bit converter assuming a 1 MHz MPU clock frequency. The simulated clock pulse is 7 μs wide and can be used to indicate when to sample the serial output.

FIGURE 6 – 8-Bit SA Software (Page 1 of 3)

```
 1.000  NAM DWA12
 2.000  OPT MEM
 3.000 *
 4.000 *************************************************************************
 5.000 *                                                                       *
 6.000 *          8 BIT SUCCESSIVE APPROXIMATION A/D                           *
 7.000 *                                                                       *
 8.000 *************************************************************************
 9.000 *
10.000 *
11.000  ORG 0
12.000 ANS RMB 1          FINAL ANSWER MEMORY LOCATION
13.000 POINTR RMB 1       TEMP MEMORY LOCATION
14.000 *
15.000 *
16.000 *
17.000  ORG $4004         PIA MEMORY ADDRESSES
18.000 PIA1AD RMB 1       A SIDE, DATA REGISTER
19.000 PIA1AC RMB 1       A SIDE, CONTROL REGISTER
20.000 PIA1BD RMB 1       B SIDE, DATA REGISTER
21.000 PIA1BC RMB 1       B SIDE, CONTROL REGISTER
22.000 *
23.000 *                  PIA1AD USED FOR DIGITAL OUTPUT TO DAC
24.000 *                  PIA1BD USED FOR A/D CONTROL
25.000 *
26.000 *
27.000 *
28.000 *******************PIA1BD PIN CONNECTIONS*******************
29.000 *************************************************************
30.000 * PB7   *  PB6 *  PB5 *  PB4 *  PB3 *  PB2 *  PB1 *  PB0 *
31.000 *************************************************************
32.000 * COMP  *  NC  *  SC  *  CF  *  SO  *  NC  * CYCLE * NC  *
33.000 *************************************************************
34.000 *                        *
35.000 *
```

FIGURE 6 — 8-Bit SA Software (Page 2 of 3)

```
36.000 *
37.000 * COMP-COMPARATOR,SC-SIMULATED CLOCK,SO-SERIAL OUTPUT
38.000 * CF-CONVERSION FINISHED, NC-NO CONNECTION
39.000 *
40.000 *
41.000 *
42.000  ORG $0A00          BEGINNING ADDRESS
43.000 *
44.000 *                              **PIA ASSEMBLY**
45.000  CLR PIA1AC
46.000  CLR PIA1BC
47.000  LDA A #$7C
48.000  STA A PIA1BD
49.000  LDA A #$0FF
50.000  STA A PIA1AD      A SIDE ALL OUTPUTS
51.000  LDA A #$04
52.000  STA A PIA1AC
53.000  STA A PIA1BC
54.000 *
55.000 RSTART LDA A #$10
56.000  STA A PIA1BD      SET CONVERSION FINISHED
57.000 *
58.000 *
59.000 *
60.000 *                              **CYCLE TEST**
61.000 *
62.000 *
63.000 CYCLE LDA A PIA1BD
64.000  AND A #$02
65.000  BEQ CYCLE
66.000 *
67.000 *
68.000  CLR PIA1AD
69.000  CLR POINTR
70.000 *
71.000 *
72.000 *
73.000  CLR PIA1BD        RESET CONVERSION FINISHED
74.000  SEC
75.000 *
76.000 *
77.000 *
78.000 *
79.000 CONVRT ROR POINTR
80.000  BCS RSTART
81.000  LDA A PIA1AD      RECALL PREVIOUS DIGITAL OUTPUT
82.000  ADD A POINTR
83.000  STA A PIA1AD      SET NEW DIGITAL OUTPUT
84.000 *
85.000 *                              **DELAY FOR COMPARATOR**
86.000 *
87.000  NOP
88.000  NOP
89.000  NOP
90.000  NOP
91.000  LDA A PIA1BD      COMPARATOR TEST
92.000  BMI YES
93.000 *
94.000 *                              **LOW COMPARATOR LOOP**
95.000  LDA A PIA1AD
96.000  SUB A POINTR
```

```
 97.000  LDA B #$20        SERIAL OUT OF "0", CLOCK SET
 98.000  STA B PIA1BD
 99.000  CLR B             CLOCK RESET
100.000  STA B PIA1BD
101.000  BRA END
102.000 *
103.000 *                          **HIGH COMPARATOR LOOP**
104.000 YES LDA A PIA1AD
105.000  NOP
106.000  NOP               DELAY
107.000  NOP
108.000  LDA B #$28        SERIAL OUTPUT OF "1", CLOCK SET
109.000  STA B PIA1BD
110.000  LDA B #$08        CLOCK RESET
111.000  STA B PIA1BD
112.000 *
113.000 END STA A PIA1AD
114.000  STA A ANS
115.000  BRA CONVRT
116.000 *
117.000 *
118.000 *
119.000 *
120.000 *
121.000 *
122.000  MON
```

FIGURE 6 – 8-Bit SA Software (Page 3 of 3)



FIGURE 7 – 8-Bit SA Hardware

10-Bit SA Program

Figures 8 and 9 show the MPU software and external hardware for a 10-bit successive approximation A/D using the MC3410 DAC. The operation of this A/D is very similar to that of the 8-bit A/D. Both the A and B halves of a PIA are required for the DAC output while the control lines (comparator, conversion finished, etc.) are also identical to that of the 8-bit A/D previously discussed. The pointer for indicating which bit is currently under test is contained in two memory locations, PONTR1 and PONTR2. The pointer is initialized in lines 63 and 64 and as before, it is continuously shifted to the left as each bit is tested. Lines 72 through 77 and lines 89 through 101 operate on both halves of the PIA, "setting" and "resetting" the DAC bits under test. The final answer is stored in the two PIA memory locations as well as two internal memory locations (ANS1 and ANS2).

By using the appropriate DAC and changing line 63 of the software program, the 10-bit SA D/A can be modified for 9-16 bit A/D operation.

FIGURE 8 – 10-Bit SA Software (Page 1 of 3)

```
 1.000  NAM DWA40
 2.000  OPT MEM
 3.000 *
 4.000 ************************************************************
 5.000 *                                                          *
 6.000 *          10 BIT SUCCESSIVE APPROXIMATION A/D             *
 7.000 *                                                          *
 8.000 ************************************************************
 9.000 *
10.000 *
11.000  ORG 0
12.000 ANS1 RMB 1          FINAL ANSWER LOCATION (MSB)
13.000 ANS2 RMB 1          FINAL ANSWER LOCATION (LSB)
14.000 PONTR1 RMB 1        POINTER FOR BIT UNDER TEST
15.000 PONTR2 RMB 1        POINTER FOR BIT UNDER TEST
16.000 *
17.000 *
18.000  ORG $4006         PIA MEMORY ADDRESSES
19.000 PIA1BD RMB 1        B SIDE, DATA REGISTER
20.000 PIA1BC RMB 1        B SIDE, CONTROL REGISTER
21.000 PIA2AD RMB 1        A SIDE, DATA REGISTER
21.500 PIA2AC RMB 1        A SIDE, CONTROL REGISTER
22.000 PIA2BD RMB 1        B SIDE, DATA REGISTER
23.000 PIA2BC RMB 1        B SIDE, CONTROL REGISTER
24.000 *
25.000 *                   PIA1AD USED FOR DIGITAL OUTPUT TO DAC
26.000 *                   PIA1BD USED FOR A/D CONTROL
27.000 *
28.000 *
29.000 *
30.000 ********************PIA1BD PIN CONNECTIONS********************
31.000 ************************************************************
32.000 * PB7   *  PB6 *  PB5 *  PB4 *  PB3 *  PB2 *  PB1 *  PB0 *
33.000 ************************************************************
34.000 * COMP  *  NC  *  SC  *  CF  *  SO  *  NC  * CYCLE * NC  *
35.000 ************************************************************
36.000 *                              *
37.000 *
38.000 *
39.000 * COMP-COMPARATOR,SC-SIMULATED CLOCK,SO-SERIAL OUTPUT
40.000 * CF-CONVERSION FINISHED, NC-NO CONNECTION
41.000 *
42.000 *
43.000 *
44.000 *
45.000 *
46.000  ORG $0A00            BEGINNING OF PROGRAM
47.000 *                           *PIA ASSEMBLY*
48.000  CLR PIA1BC
49.000  CLR PIA2AC
```

FIGURE 8 – 10-Bit SA Software (Page 2 of 3)

```
 50.000  CLR PIA2BC
 51.000  LDA A #$7C
 52.000  STA A PIA1BD
 53.000  LDA A #$0FF
 54.000  STA A PIA2AD
 55.000  STA A PIA2BD
 56.000  LDA A #$04
 57.000  STA A PIA1BC
 58.000  STA A PIA2AC
 59.000  STA A PIA2BC
 60.000 *
 61.000 RESTART LDA A #$10
 62.000  STA A PIA1BD          SET CONVERSION FINISHED
 63.000  CLR PONTR1
 64.000  CLR PONTR2
 65.000 *
 66.000 *
 67.000 *
 68.000 *                          *CYCLE TEST*
 69.000 *
 70.000 *
 71.000 CYCLE LDA A PIA1BD
 72.000  AND A #$02
 73.000  BEQ CYCLE
 74.000 *                          *RESET DAC INPUTS*
 75.000  CLR PIA2AD
 76.000  CLR PIA2BD
 77.000 *
 78.000 *
 79.000 *
 80.000  CLR PIA1BD          RESET CONVERSION FINISHED
 81.000  LDA A #$04
 82.000  STA A PONTR1
 83.000 *
 84.000 *
 85.000 *
 86.000 *
 87.000 CONVRT ROR PONTR1
 88.000  ROR PONTR2
 89.000  BCS RESTART
 90.000  LDA A PIA2AD        RECALL PREVIOUS DIGITAL OUTPUT(8 LSB)
 91.000  ADD A PONTR1
 92.000  STA A PIA2AD        SET NEW DIGITAL OUTPUT
 93.000  LDA A PIA2BD        RECALL PREVIOUS DIGITAL OUTPUT(2 MSB)
 94.000  ADD A PONTR2
 95.000  STA A PIA2BD        SET NEW DIGITAL OUTPUT
 96.000 *
 97.000 *                        *DELAY FOR COMPARATOR*
 98.000 *
 99.000  NOP
100.000  NOP
101.000  NOP
102.000  NOP
103.000  LDA A PIA1BD        COMPARATOR TEST
104.000  BMI YES
105.000 *
106.000 *
107.000 *                        *LOW COMPARATOR LOOP*
108.000  LDA A PIA2AD
109.000  SUB A PONTR1
110.000  STA A PIA2AD
111.000  STA A ANS1
112.000  LDA A PIA2BD
```

```
113.000  SUB A PONTR2
114.000  STA A PIA2BD
115.000  STA A ANS2
116.000  LDA B #$20          SERIAL OUTPUT (CLOCK ONLY)
117.000  STA B PIA1BD
118.000  CLR B               CLOCK RESET
119.000  STA B PIA1BD
120.000  BRA END
121.000 *
122.000 *
123.000 *                          *HIGH COMPARATOR LOOP*
124.000 YES LDA A #$05      TIME EQUALIZATION
125.000 DELAY DEC A
126.000  BNE DELAY
127.000  LDA B #$28          SERIAL OUTPUT
128.000  STA B PIA1BD
129.000  LDA B #$08          CLOCK RESET
130.000  STA B PIA1BD
131.000  NOP
132.000  NOP
133.000 *
134.000 END BRA CONVRT
135.000 *
136.000 *
137.000 *
138.000  MON
```

FIGURE 8 – 10-Bit SA Software (Page 3 of 3)



FIGURE 9 – 10-Bit SA Hardware

External SA System

The third successive approximation program, shown in Figures 10 and 11, uses an MC1408 DAC with the MC14549 CMOS SAR for a convert-on-command A/D system. This system is controlled by the MPU through the CA1 and CA2 PIA pins to start a conversion and store the results of this conversion in memory when the conversion is finished. The 8-bit data word from the A/D is brought in to the MPU system through PIA1AD. The advantages of this A/D system are that a minimum number of software instructions are required, a higher speed conversion is possible, and the MPU may be performing other tasks during the conversion. The disadvantage is a higher parts count and increased cost.

The program for this A/D, shown in Figure 11, is written as a subroutine of a larger program. This larger program is simulated with the instructions of lines 28 through 31. The subroutine starts in line 34, unmasking the interrupt input on CA1 and setting CA2 high. (For additional information on use of the CA1 and CA2 lines, see the MC6820 data sheet.) CA2 initiates the conversion. Line 35 is a dummy read statement necessary to clear the data register of the interrupt bit associated with the CA1 input line. Then a wait for interrupt instruction stores the stack in anticipation of the A/D conversion being completed. When the conversion is finished the CA1 line is toggled by the EOC output of the MC14549 and the program goes to line 43 where CA1 is masked and CA2 is set low, thus stopping any further conversion sequences by the A/D. The digital results are loaded into the A accumulator through PIA-A and stored in memory location TEMP. Then the MPU returns from the interrupt and finally returns from the subroutine.

The entire sequence requires 60 μs plus the conversion time of the A/D.



FIGURE 10 – 8-Bit SA Using External Hardware

FIGURE 11 – 8-Bit External SA Software (Page 1 of 2)

```
1.000  NAM DWA3
2.000  OPT OT
3.000 *
4.000 ********************************************************************
5.000 *                    8 BIT BINARY SUCCESSIVE                       *
6.000 *                    APPROXIMATION HARDWARE                        *
7.000 ********************************************************************
8.000 *
9.000  ORG $0100
```

```
10.000 TEMP RMB 1                    8 BIT BINARY DATA
11.000 *
12.000 *
13.000  ORG $4004
14.000 PIA1AD RMB 1                  DATA REGISTER
15.000 PIA1AC RMB 1                  CONTROL REGISTER
16.000 *
17.000 *
18.000 *
19.000 *
20.000  ORG $0300
21.000  CLR PIA1AC                   PIA ASSEMBLY
22.000  CLR PIA1AD
23.000  LDA A #$3C
24.000  STA A PIA1AC
25.000  LDS #$0020
26.000 *
27.000 *
28.000  NOP
29.000  JSR CONVRT
30.000 END NOP
31.000  BRA END
32.000 *
33.000 *                          CONVERSION SUBROUTINE
34.000 CONVRT LDA A #$3F             CA1 UNMASKED,POS EDGE--CA2 HIGH
35.000  LDA B PIA1AD
36.000  STA A PIA1AC
37.000  WAI
38.000  RTS
39.000 *
40.000 *
41.000 *
42.000 *                           INTERRUPT PROGRAM
43.000 INTRPT LDA A #$36             CA1 MASKED-CA2 LOW
44.000  STA A PIA1AC
45.000  LDA A PIA1AD
46.000  STA A TEMP
47.000  RTI
48.000 *
49.000 *
50.000 *
51.000  MON
```

FIGURE 11 – 8-Bit External SA Software (Page 2 of 2)

DUAL RAMP TECHNIQUES

General

Another commonly used method for A/D conversion is the dual ramp or dual slope technique. This approach has a longer conversion time than that of the successive approximation method. The conversion time period is also variable and input voltage dependent. However, this method yields an A/D converter of high accuracy and low cost.

As the name implies the dual ramp method consists of two ramp periods for each conversion cycle. Figure 12 shows the basic waveforms for the dual ramp A/D. The ratio in time of the ramp lengths provides a value representing the difference between a reference and an unknown voltage. During time period T1, the input unknown is integrated for a fixed time period (fixed number of clock cycles). The integrator voltage increases from the reference level to a voltage which is proportional to the input voltage. At the end of this time period a reference voltage is applied to the input of the integrator causing the integrator output voltage to decrease until the reference level is again reached. The number of clock cycles that are required to bring the integrator output voltage back to the reference level is proportional to the input unknown voltage.

CA2 output of PIA1BC control register. Then the conversion is restarted, this time with a positive input polarity. The polarity detection instruction is found in line 131. If after the offset count subtraction in lines 129 and 130 the condition code carry bit is "set", the MC1405 has a negative input voltage. This occurs when the negative input subtracts from instead of adding to the offset current in the MC1405 and does not allow the ramp down time period to reach at least a value of 100_{10} counts. If the carry bit has been "set" then the program branches to line 236 where the CA2 line is toggled. Also due to the difference in a positive polarity conversion and a negative polarity conversion a short delay loop has been added in lines 238 and 239 to improve accuracy at very small input voltages.

The entire 3½-digit A/D requires 296 memory locations but can be reduced if the BCD-to-7 segment decoding is performed external to the MPU system. With a 1 MHz MPU clock frequency this program has a full scale conversion time of 60 ms.



FIGURE 16 – 3½-Digit Dual Ramp A/D Flow Diagram

FIGURE 17 – 3½-Digit Dual Ramp Software (Page 1 of 5)

```
 1.000  NAM DWA25
 2.000  OPT MEM
 3.000 *
 4.000 *
 5.000 *        ********************************************************
 6.000 *        *                                                      *
 7.000 *        *              3 1/2 DIGIT A/D                         *
 8.000 *        *                                                      *
 9.000 *        ********************************************************
10.000 *
11.000 *
12.000 * THIS CONVERTER USES A MC1405 IN CONJUNCTION WITH THE
13.000 * MC6800 MPU TO PRODUCE A 3 1/2 DIGIT A/D.  THE
14.000 * DUAL RAMP METHOD OF A/D CONVERSION IS USED.
15.000 *
16.000 *   THE INPUTS TO THE MPU CONSIST OF
17.000 *
18.000 *        CYCLE SWITCH -LOCATED AT PIA1BD (PB1)
19.000 *        COMPARATOR   = LOCATED AT PIA1BD (PB7)
20.000 *
21.000 *   THE OUTPUTS FROM THE MPU CONSIST OF
22.000 *
23.000 *        RAMP CONTROL- LOCATED AT PIA1BD (PB2)
24.000 *        CONVERSION FINISHED = LOCATED AT PIA1BD (PB4)
25.000 *        OVERRANGE   - LOCATED AT PIA1BD (PB3)
26.000 *        POLARITY    - LOCATED AT PIA1BD (CA2)
27.000 *
28.000 *        7 SEGMENT OUTPUT
29.000 *          TENS  - PIA1AD
30.000 *          HUNDREDS - PIA2AD
31.000 *          THOUSANDS - PIA2BD
32.000 *          TENS OF THOUSANDS OR HALF DIGIT - PIA2BD (PB7)
33.000 *
34.000 *    THE BINARY ANSWER IS STORED AT MSB AND LSB
35.000 *
36.000 *    THE BCD ANSWER IS STORED AT UNTTEN,HNDTHD,TENTSD
37.000 *
38.000 *    THE ANALOG INPUT FOR THE MC1405 MUST HAVE A 2 VOLT
39.000 *    MAXIMUM WHILE THE AUTOPOLARITY OUTPUT FROM THE MPU
40.000 *    MAY BE USED TO TOGGLE A RELAY TO PROVIDE NEGATIVE
41.000 *    INPUT CAPABILITY FOR THE A/D
42.000 *
43.000 *
44.000 *
45.000  ORG $0000
46.000 MSB RMB 1
47.000 LSB RMB 1
48.000 INDEX RMB 2
49.000 MSBTEM RMB 1          TEMP STORAGE OF BINARY ANSWER
50.000 LSBTEM RMB 1
51.000 *
52.000 *
53.000 *
54.000  ORG $0010
55.000 UNTTEN RMB 1
56.000 HNDTHD RMB 1
57.000 *
58.000 *
59.000  ORG $4004
60.000 PIA1AD RMB 1          A SIDE DATA REGISTER
```

FIGURE 17 — 3½-Digit Dual Ramp Software (Page 2 of 5)

```
 61.000 PIA1AC RMB 1          A SIDE CONTROL REGISTER
 62.000 PIA1BD RMB 1          B SIDE DATA REGISTER
 63.000 PIA1BC RMB 1          B SIDE CONTROL REGISTER
 64.000 PIA2AD RMB 1          A SIDE DATA REGISTER
 65.000 PIA2AC RMB 1          A SIDE CONTROL REGISTER
 66.000 PIA2BD RMB 1          B SIDE DATA REGISTER
 67.000 PIA2BC RMB 1          B SIDE CONTROL REGISTER
 68.000 *
 69.000 *
 70.000  ORG $0A00
 71.000 *
 72.000 *                                    **PIA ASSEMBLY**
 73.000  CLR PIA1AC
 74.000  CLR PIA1BC
 75.000  CLR PIA2AC
 76.000  CLR PIA2BC
 77.000  LDA A #$7C
 78.000  STA A PIA1BD
 79.000  LDA A #$0FF
 80.000  STA A PIA1AD
 81.000  STA A PIA2AD
 82.000  STA A PIA2BD
 83.000  LDA A #$34       SETS PIA CONTROL REGISTER BIT 3 HIGH
 84.000  STA A PIA1AC
 85.000  STA A PIA1BC
 86.000  STA A PIA2AC
 87.000  STA A PIA2BC
 88.000 *
 89.000  LDA A #$0C           FIRST TWO HEX DIGITS OF LOOK-UP
 90.000  STA A INDEX                TABLE ADDRESSES
 91.000 *                  ******************
 92.000 *                  *   BASIC A/D    *
 93.000 *                  ******************
 94.000 *
 95.000 *                                    INITIALIZATION
 96.000  LDA A #$04
 97.000  STA A PIA1BD   RC HIGH
 98.000 START LDA A PIA1BD  COMPARATOR TEST
 99.000  BMI START
100.000 CYCLE1 LDA A #$14
101.000  STA A PIA1BD   CONVERSION READY AND RC HIGH
102.000 *
103.000 *
104.000 *                                   **CYCLE TEST**
105.000 CYCLE LDA A PIA1BD
106.000  AND A #$02
107.000  BEQ CYCLE
108.000 RESTAR LDX #$07D0
109.000  CLR PIA1BD RESET OVERRANGE, CONVERSION FINISHED AND SET RC LOW
110.000 COMP LDA A PIA1BD
111.000  BPL COMP
112.000 *                                   **RAMP UP TIMING CYCLE**
113.000 RAMPUP LDA B #$04
114.000  DEX
115.000  BNE RAMPUP
116.000 *
117.000 *                                   **RAMP DOWN TIMING CYCLE**
118.000 *
119.000 *
120.000 RAMPDN STA B PIA1BD     RC HIGH
```

```
121.000  INX
122.000  CPX #0000            DUMMY STATEMENT FOR TIME DELAY
123.000  LDA A PIA1BD  COMPARATOR TEST
124.000  BMI RAMPDN
125.000 *
126.000  STX MSB
127.000  LDA A MSB+1
128.000  LDA B MSB
129.000  SUB A #$64
130.000  SBC B #$00
131.000  BCS POLRY1
132.000  STA A MSB+1
133.000  STA B MSB
134.000  STA A MSBTEM+1
135.000  STA B MSBTEM
136.000 *
137.000 *
138.000 *
139.000 *                    ******************
140.000 *                    * BINARY TO BCD *
141.000 *                    *   CONVERTER    *
142.000 *                    ******************
143.000 *
144.000  CLR UNTTEN
145.000  CLR HNDTHD
146.000  LDX #$0010
147.000 BEGIN LDA A UNTTEN
148.000  TAB
149.000  AND A #$0F
150.000  SUB A #$05
151.000  BMI AT
152.000  ADD B #$03
153.000 AT TBA
154.000  AND A #$0F0
155.000  SUB A #$50
156.000  BMI BT
157.000  ADD B #$30
158.000 BT STA B UNTTEN
159.000 *
160.000  LDA A HNDTHD
161.000  TAB
162.000  AND A #$0F
163.000  SUB A #$05
164.000  BMI CT
165.000  ADD B #$03
166.000 CT TBA
167.000  AND A #$0F0
168.000  SUB A #$50
169.000  BMI DT
170.000  ADD B #$30
171.000 DT STA B HNDTHD
172.000 *
173.000 *
174.000 *
175.000  ASL LSBTEM
176.000  ROL MSBTEM
177.000  ROL UNTTEN
178.000  ROL HNDTHD
179.000  DEX
180.000  BNE BEGIN
```

FIGURE 17 — 3½-Digit Dual Ramp Software (Page 3 of 5)

4½-Digit Dual Ramp Program

The microprocessor software for a 4½-digit dual ramp A/D is shown in Figure 19. This program in an extension of the 3½-digit A/D just discussed and has a full scale input voltage of 1.9999 volts. Due to the addition of the extra digit, a fourth PIA port for the 7-segment display is required. The PIA port configuration used for ramp control, comparator, etc. is identical to that used in the 3½-digit A/D.

The addition of the extra digit also implies a longer ramp up time period which is produced by increasing the initialization of the index register in line 115. This longer ramp up time period also requires the change of the extra count subtraction statements of lines 137 and 138 to maintain the extra count subtraction of 10% ramp up time. Also, the longer ramp up time period will require a larger integration capacitor to prevent saturation of the MC1405 integrator. This is of course, assuming the same MPU clock frequency. The remainder of the A/D external hardware is unchanged except for the addition of the fourth full digital display. Figure 18a can be used for the 4½-digit A/D without modification, and Figure 18b can be used with only the addition of another digit.

The software for the binary-to-BCD converter remains the same for the 4½-digit A/D since it is capable of handling up to 16 bits. The conversion routine for BCD-to-7 segment code must be modified to handle the extra digit although the same basic technique is retained.

FIGURE 19 – 4½-Digit Dual Ramp Software (Page 1 of 5)

```
 1.000  NAM DWA 30
 2.000  OPT MEM
 3.000 *
 4.000 *
 5.000 *        ************************************************
 6.000 *        *                                              *
 7.000 *        *            4 1/2 DIGIT A/D                   *
 8.000 *        *                                              *
 9.000 *        ************************************************
10.000 *
11.000 *
12.000 * THIS CONVERTER USES A MC1405 IN CONJUNCTION WITH THE
13.000 * MC6800 MPU TO PRODUCE A 4 1/2 DIGIT A/D.  THE
14.000 * DUAL RAMP METHOD OF A/D CONVERSION IS USED.
15.000 *
16.000 *   THE INPUTS TO THE MPU CONSIST OF
17.000 *
18.000 *        CYCLE SWITCH -LOCATED AT PIA1BD (PB1)
19.000 *        COMPARATOR   - LOCATED AT PIA1BD (PB7)
20.000 *
21.000 *   THE OUTPUTS FROM THE MPU CONSIST OF
22.000 *
23.000 *        RAMP CONTROL- LOCATED AT PIA1BD (PB0)
24.000 *        CONVERSION FINISHED - LOCATED AT PIA3BD (PB1)
25.000 *        OVERRANGE   - LOCATED AT PIA1BD (PB2)
26.000 *        POLARITY    - LOCATED AT PIA1BD (PB6)
27.000 *
28.000 *        7 SEGMENT OUTPUT
29.000 *          TENS  - PIA2BD
30.000 *          HUNDREDS - PIA3AD
31.000 *          THOUSANDS - PIA3BD
32.000 *          TENS OF THOUSANDS OR HALF DIGIT -PIA3BD (PB7)
33.000 *
34.000 *    THE BINARY ANSWER IS STORED AT MSB AND LSB
35.000 *
36.000 *    THE BCD ANSWER IS STORED AT UNTTEN,HNDTHD,TENTSD
37.000 *
38.000 *    THE ANALOG INPUT FOR THE MC1405 MUST HAVE A 2 VOLT
39.000 *    MAXIMUM WHILE THE AUTOPOLARITY OUTPUT FROM THE MPU
40.000 *    MAY BE USED TO TOGGLE A RELAY TO PROVIDE NEGATIVE
41.000 *    INPUT CAPABILITY FOR THE A/D
42.000 *
43.000 *
44.000 *
45.000  ORG $0000
```

```
 46.000 MSB RMB 1
 47.000 LSB RMB 1
 48.000 INDEX RMB 2
 49.000 MSBTEM RMB 1            TEMP STORAGE OF BINARY ANSWER
 50.000 LSBTEM RMB 1
 51.000 *
 52.000 *
 53.000 *
 54.000  ORG $0010
 55.000 UNTTEN RMB 1
 56.000 HNDTHD RMB 1
 57.000 TENTSD RMB 1
 58.000 *
 59.000 *
 60.000  ORG $4006
 61.000 PIA1BD RMB 1          B SIDE, DATA REGISTER
 62.000 PIA1BC RMB 1          B SIDE, CONTROL REGISTER
 63.000 PIA2AD RMB 1          A SIDE, DATA REGISTER
 64.000 PIA2AC RMB 1          A SIDE, CONTROL REGISTER
 65.000 PIA2BD RMB 1          B SIDE, DATA REGISTER
 66.000 PIA2BC RMB 1          B SIDE, CONTROL REGISTER
 67.000  ORG $4010
 68.000 PIA3AD RMB 1          A SIDE, DATA REGISTER
 69.000 PIA3AC RMB 1          A SIDE, CONTROL REGISTER
 70.000 PIA3BD RMB 1          B SIDE, DATA REGISTER
 71.000 PIA3BC RMB 1          B SIDE, CONTROL REGISTER
 72.000 *
 73.000 *
 74.000 *
 75.000 *                                  PIA ASSEMBLY
 76.000  ORG $0A00
 77.000  CLR PIA1BC
 78.000  CLR PIA2AC
 79.000  CLR PIA2BC
 80.000  CLR PIA3AC
 81.000  CLR PIA3BC
 82.000  LDA A #$4D
 83.000  STA A PIA1BD
 84.000  LDA A #$0FF     REMAINING PIA'S ALL OUTPUTS
 85.000  STA A PIA2AD
 86.000  STA A PIA2BD
 87.000  STA A PIA3AD
 88.000  STA A PIA3BD
 89.000  LDA A #$34       SETS PIA CONTROL REGISTER BIT 3 HIGH
 90.000  STA A PIA1BC
 91.000  STA A PIA2AC
 92.000  STA A PIA2BC
 93.000  STA A PIA3AC
 94.000  STA A PIA3BC
 95.000 *
 96.000  LDA A #$0C          FIRST TWO HEX DIGITS OF LOOK-UP
 97.000  STA A INDEX                  TABLE ADDRESSES
 98.000 *                  ****************
 99.000 *                  *   BASIC A/D  *
100.000 *                  ****************
101.000 *
102.000 *                                  INITIALIZATION
103.000  LDA A #$04
104.000  STA A PIA1BD   RC HIGH
105.000 START LDA A PIA1BD  COMPARATOR TEST
```

FIGURE 19 – 4½-Digit Dual Ramp Software (Page 2 of 5)

APPENDIX B

BINARY-TO-BCD CONVERSION

A standard technique for binary-to-BCD conversion is that of the Add 3 algorithm. Figures B1 and B2 show a flow diagram and example of this algorithm. The technique requires a register containing the N-bit binary number and enough 4-bit BCD registers to contain the maximum equivalent BCD number for the initial binary number. The conversion starts by checking each BCD register for a value of 5 or greater. If this condition exists in one or all of these registers (initially this condition cannot exist), then a 3 is added to those registers where this condition exists. Next the registers are shifted left with the carry out of the previous register being the carry in to the next register. Again each BCD register is checked for values of 5 or greater. This sequence continues until the registers have been shifted N times, where N is the number of bits in the initial binary word. The BCD registers then contain the resulting BCD equivalent to the initial binary word. The example in Figure B2 starts with an 8-bit binary word consisting of all "1's." This word is converted to the BCD equivalent of 255 by this technique. After 8 shifts the last binary bit has been shifted out of the binary register and the hundreds, tens, and units registers contain a 255.

Figure B3 shows an MC6800 software routine for performing this technique of binary to BCD conversion. The initial binary number is a 16-bit number and occupies memory locations MSB and LSB; this binary number is converted to the equivalent BCD number in memory locations TENTSD, HNDTHD and UNTTEN. Each of these memory locations contains two BCD digits. Eighty-three memory locations are required for program storage with a maximum conversion taking 1.8 ms.



FIGURE B1 – Binary to BCD Conversion Flow Diagram

Hundreds	Tens	Units	8-Bit Binary	
			1 1 1 1 1 1 1 1	
		1	1 1 1 1 1 1 1	Shift
		1 1	1 1 1 1 1 1	Shift
		1 1 1	1 1 1 1 1	Shift
		1 0 1 0	1 1 1 1 1	Add 3 to Units
	1	0 1 0 1	1 1 1 1	Shift
	1	1 0 0 0	1 1 1 1	Add 3 to Units
	1 1	0 0 0 1	1 1 1	Shift
	1 1 0	0 0 1 1	1 1	Shift
	1 0 0 1	0 0 1 1	1 1	Add 3 to Tens
1	0 0 1 0	0 1 1 1	1	Shift
1	0 0 1 0	1 0 1 0	1	Add 3 to Units
1 0	0 1 0 1	0 1 0 1		Shift
2	5	5		Total Shifts 8

FIGURE B2 – Binary to BCD Conversion

FIGURE B3 – Binary-to-BCD Conversion Software (Page 1 of 2)

```
 1.000  NAM DWA21
 2.000  OPT MEM
 3.000 *
 4.000 *************************************************************
 5.000 *                                                           *
 6.000 *                BINARY TO BCD CONVERSION                   *
 7.000 *                   ADD 3 ALGORITHM                         *
 8.000 *                      16 BIT                               *
 9.000 *************************************************************
10.000 *
11.000  ORG 0              INITIAL BINARY NUMBER
12.000 MSB RMB 1             MOST SIGNIFICANT 8 BITS
13.000 LSB RMB 1             LEAST SIGNIFICANT 8 BITS
14.000 *
15.000 *
16.000 *
17.000  ORG $0010          BCD RESULTS
18.000 UNTTEN RMB 1          UNITS AND TENS DIGITS
19.000 HNDTHD RMB 1          HUNDREDS AND THOUSANDS
20.000 TENTSD RMB 1          TENS OF THOUSANDS DIGIT
21.000 *
22.000 *
23.000 *
```

```
24.000  ORG $0F00        **BEGINNING OF PROGRAM**
25.000  CLR UNTTEN
26.000  CLR HNDTHD
27.000  CLR TENTSD
28.000  LDX #$0010
29.000 BEGIN LDA A UNTTEN      UNITS COMPARISON
30.000  TAB
31.000  AND A #$0F
32.000  SUB A #$05
33.000  BMI AT
34.000  ADD B #$03
35.000 AT TBA                  TENS COMPARISON
36.000  AND A #$0F0
37.000  SUB A #$50
38.000  BMI BT
39.000  ADD B #$30
40.000 BT STA B UNTTEN
41.000 *
42.000  LDA A HNDTHD           HUNDREDS COMPARISON
43.000  TAB
44.000  AND A #$0F
45.000  SUB A #$05
46.000  BMI CT
47.000  ADD B #$03
48.000 CT TBA
49.000  AND A #$0F0
50.000  SUB A #$50
51.000  BMI DT
52.000  ADD B #$30
53.000 DT STA B HNDTHD
54.000 *
55.000  LDA A TENTSD           TENS OF THOUSANDS COMPARISON
56.000  TAB
57.000  SUB A #$05
58.000  BMI ET
59.000  ADD B #$03
60.000 ET STA B TENTSD
61.000 *
62.000 *
63.000  ASL LSB
64.000  ROL MSB
65.000  ROL UNTTEN
66.000  ROL HNDTHD
67.000  ROL TENTSD
68.000  DEX
69.000  BNE BEGIN              END OF CONVERSION CHECK
70.000 *
71.000 *
72.000 *
73.000 *
74.000  END
75.000  MON
```

FIGURE B3 — Binary-to-BCD Conversion Software (Page 2 of 2)

MOTOROLA Semiconductor Products Inc.

Printed in Switzerland